

Monads

- 22 Invent your first monad
- 23 More examples of monads
- 24 Monads and their laws
- 25 Program transformations and monads
- 26 Monads as a general programming technique
- 27 Monads and ML Functors

Exception-returning style, state-passing style, and continuation-passing style of the previous part are all special cases of *monads*

Monads are thus a technical device that factor out commonalities between many program transformations ...

... but this is just one possible viewpoint. Besides that, they can be used

- To structure denotational semantics and make them easy to extend with new language features. (E. Moggi, 1989.)
- As a powerful programming techniques in pure functional languages, primary in Haskell. (P. Wadler, 1992).

- 22 Invent your first monad
- 23 More examples of monads
- 24 Monads and their laws
- 25 Program transformations and monads
- 26 Monads as a general programming technique
- 27 Monads and ML Functors

Invent your first monad

Probably the best way to understand monads is to define one. Or better, arrive to a point where you realize that you need one (even if you do not know that it is a monad).

Many of the problems that monads try to solve are related to the issue of side effects. So we'll start with them.

Side Effects: Debugging Pure Functions



Input: We have functions f and g that both map floats to floats.

$f, g : \text{float} \rightarrow \text{float}$

Goal: Modify these functions to output their calls for debugging purposes

If we do not admit side effects, then the modified version f' and g' must return the output

$f', g' : \text{float} \rightarrow \text{float} * \text{string}$



We can think of these as 'debuggable' functions.

Binding


Problem: How to debug the composition of two 'debuggable' functions?

Intuition: We want the composition to have type `float -> float * string`
but types no longer work!

Solution: Use concatenation for the debug messages and add some plumbing

```
let (y,s) = g' x in  
let (z,t) = f' y in (z,s^t)    (where ^ denotes string concatenation)
```

Diagrammatically:



The bind function

Plumbing is ok ... once. To do it uniformly we need a higher-order function doing the plumbing for us. We need a function `bind` that upgrades `f'` so that it can be plugged in the output of `g'`. That is, we would like:

```
bind f' : (float*string) -> (float*string)
```

which implies that

```
bind : (float -> (float*string)) -> ((float*string) -> (float*string))
```

`bind` must

- 1 apply `f'` to the correct part of `g'` `x` and
- 2 concatenate the string returned by `g'` with the string returned by `f'`.

Exercise

Write the function `bind`.

```
# let bind f' (gx,gs) = let (fx,fs) = f' gx in (fx,gs^fs)
val bind : ('a -> 'b * string) -> 'a * string -> 'b * string = <fun>
```

The return function

Given two debuggable functions, f' and g' , now they can be composed by `bind`

`(bind f')` . `g'` (where “.” is Haskell's infix composition).

Write this composition as $f' \circ g'$.

We look for a “debuggable” identity function `return` such that for every debuggable function f one has $\text{return} \circ f = f \circ \text{return} = f$.

Exercise

Define `return`.

```
# let return x = (x, "");;  
val return : 'a -> 'a * string = <fun>
```

In Haskell (from now on we switch to this language):

```
Prelude> let return x = (x, "")  
Prelude> :type return  
return :: t -> (t, [Char])  --t is a schema variable, String = Char list
```

In summary, the function `return` lifts the result of a function into the result of a “debuggable” function.

The lift function

The return allows us to “lift” any *function* into a debuggable one:

```
lift f = return . f
```

that is `lift f x = (f x, "")`

The lifted version does much the same as the original function and, quite reasonably, it produces the empty string as a side effect.

Exercise

Show that `lift f ∘ lift g = lift (f.g)`

Summary

The functions, `bind` and `return`, allow us to compose debuggable functions in a straightforward way, and compose ordinary functions with debuggable functions in a natural way.

We just defined our first monad
Let us see more examples

- 22 Invent your first monad
- 23 More examples of monads**
- 24 Monads and their laws
- 25 Program transformations and monads
- 26 Monads as a general programming technique
- 27 Monads and ML Functors

A Container: Multivalued Functions

Consider `sqrt` and `cbrt` that compute the square root and cube root of a real number:

```
sqrt,cbrt :: Float -> Float
```

Consider the complex version for these functions. They must return *lists* of results (two square roots and three cube roots)¹

```
sqrt',cbrt' :: Complex -> [Complex]
```

since they are *multi-valued* functions.

We can compose `sqrt` and `cbrt` to obtain the sixth root function

```
sixthrt x = sqrt (cbrt x)
```

Problem How to compose `sqrt'` and `cbrt'`?

Bind

We need a `bind` function that lifts `cbrt'` so that it can be applied to *all* the results of `sqrt'`


¹Complex should be instead written `Complex Float`, since it is a Haskell module

bind for multivalued functions

Goal:

```
bind :: (Complex -> [Complex]) -> ([Complex] -> [Complex])
```

Diagrammatically:



Exercise

Write an implementation of `bind`

Solution:

```
bind f x = concat (map f x)
```

return for multivalued functions

Again we look for an identity function for multivalued functions: it take a result of a normal function and transforms it into a result of multi-valued functions:

```
return :: a -> [a]
```

Exercise

Define return

Solution:

```
return x = [x]
```

Again

$$f \circ \text{return} = \text{return} \circ f = f$$

while $\text{lift } f = \text{return} . f$ transforms an ordinary function into a multivalued one: $\text{lift} :: (a \rightarrow b) \rightarrow a \rightarrow [b]$

**We just defined our second monad
Let us see a last one and then recap**

A more complex side effect: Random Numbers

The Haskell random function looks like this

```
random :: StdGen → (a, StdGen)
```

- To generate a random number you need a seed (of type `StdGen`)
- After you've generated the number you update the seed to a new value
- In a non-pure language the seed can be a global reference. In Haskell the new seed needs to be passed in and out explicitly.

So a function of type `a -> b` that needs random numbers must be lifted to a “randomized” function of type `a -> StdGen -> (b, StdGen)`

Exercise

- 1 Write the type of the `bind` function to compose two “randomized” functions.
- 2 Write an implementation of `bind`

A more complex side effect: Random Numbers

Solution:

- 1 $\text{bind} :: (a \rightarrow \text{StdGen} \rightarrow (b, \text{StdGen}))$
 $\rightarrow (\text{StdGen} \rightarrow (a, \text{StdGen})) \rightarrow (\text{StdGen} \rightarrow (b, \text{StdGen}))$
- 2 $\text{bind } f \ x \ \text{seed} = \text{let } (x', \text{seed}') = x \ \text{seed} \ \text{in } f \ x' \ \text{seed}'$

Exercise

Define the 'identity' randomized function. This needs to be of type

$\text{return} :: a \rightarrow (\text{StdGen} \rightarrow (a, \text{StdGen}))$

and should leave the seed unmodified.

Solution

$\text{return } x \ g = (x, g)$

Again, $\text{lift } f = \text{return} . f$ turns an ordinary function into a randomized one that leaves the seed unchanged.

While $f \circ \text{return} = \text{return} \circ f = f$ and $\text{lift } f \circ \text{lift } g = \text{lift}(f.g)$ where $f \circ g = (\text{bind } f).g$

- 22 Invent your first monad
- 23 More examples of monads
- 24 Monads and their laws**
- 25 Program transformations and monads
- 26 Monads as a general programming technique
- 27 Monads and ML Functors

Step 1: Transform a type a into the type of particular *computations* on a .

```
-- The debuggable computations on a
type Debuggable a = (a,String)
-- The multivalued computation on a
type Multivalued a = [a]
-- The randomized computations on a
type Randomized a = StdGen -> (a,StdGen)
```

Step 2: Define the “plumbing” to lift function on given types into functions on the “ m computations” on these types where “ m ” is either `Debuggable`, or `Multivalued`, or `Randomized`.

```
bind :: (a -> m b) -> (m a -> m b)
return :: a -> m a
```

with $f \circ \text{return} = \text{return} \circ f = f$ and $\text{lift } f \circ \text{lift } g = \text{lift } (f.g)$, where ‘ \circ ’ and `lift` are defined in terms of `return` and `bind`.

Monad

A *monad* is a triple formed by a type constructor `m` and two functions `bind` and `return` whose type and behavior is as described above.

Monads in Haskell

In Haskell, the bind function:

- it is written `>>=`
- it is infix
- its type is `m a -> (a -> m b) -> m b` (arguments are inverted)

This can be expressed by typeclasses:

```
class Monad m where
  -- chain computations
  (>>=) :: m a -> (a -> m b) -> m b
  -- inject
  return :: a -> m a
```

The properties of bind and return cannot be enforced, but monadic computation demands that the following equations hold

$$\begin{aligned} \text{return } x \gg= f &\equiv f x \\ m \gg= \text{return} &\equiv m \\ m \gg= (\lambda x. f x \gg= g) &\equiv (m \gg= f) \gg= g \end{aligned}$$

We already saw some of these properties:

$$\text{return } x \gg= f \equiv f x \quad (1)$$

$$m \gg= \text{return} \equiv m \quad (2)$$

$$m \gg= (\lambda x. f x \gg= g) \equiv (m \gg= f) \gg= g \quad (3)$$

Let us rewrite them in terms of our old `bind` function (with the different argument order we used before)

- 1 In (1) abstract the x then you have the *left identity*:

$$(\text{bind } f).\text{return} = f \circ \text{return} = f$$

- 2 In (2) consider $m = gx$ and abstract the x then you have the *right identity*

$$(\text{bind return}).g = \text{return} \circ g = g$$

- 3 Law (3) express *associativity* (exercise: prove it)

$$h \circ (f \circ g) = (h \circ f) \circ g$$

Writer, List and State Monads

The monads we showed are special cases of Writer, List, and State monads.
Let us see their (simplified) versions

```
-- The Writer Monad
data Writer a = Writer (a, [Char])

instance Monad Writer where
  return x          = Writer (x, [])
  Writer (x,l) >>= f = let Writer (x',l') = f x in Writer (x', l++l')
```



```
-- The List monad ([] data type is predefined)
instance Monad [] where
  return x          = [x]
  m >>= f          = concat (map f m)
```



```
-- The State Monad
data State s a = State (s -> (a,s))

instance Monad (State s) where
  return a          = State (\s -> (a,s))
  (State g) >>= f   = State (\s -> let (v,s') = g s in
                                   let State h = f v in h s')
```

QUESTION

Haven't you already seen the state monad?

Let us strip out the type constructor part:

```
return a    =  $\lambda s \rightarrow (a, s)$   
a >>= f    =  $\lambda s \rightarrow \text{let } (v, s') = a \text{ s in } (f \ v) \ s'$ 
```

It recalls somehow the transformation for the state passing style:

$$\llbracket N \rrbracket = \lambda s. (N, s)$$

$$\llbracket x \rrbracket = \lambda s. (x, s)$$

$$\llbracket \lambda x. a \rrbracket = \lambda s. (\lambda x. \llbracket a \rrbracket, s)$$

$$\llbracket \text{let } x = a \text{ in } b \rrbracket = \lambda s. \text{match } \llbracket a \rrbracket s \text{ with } (x, s') \rightarrow \llbracket b \rrbracket s'$$

$$\begin{aligned} \llbracket ab \rrbracket &= \lambda s. \text{match } \llbracket a \rrbracket s \text{ with } (x_a, s') \rightarrow \\ &\quad \text{match } \llbracket b \rrbracket s' \text{ with } (x_b, s'') \rightarrow x_a x_b s'' \end{aligned}$$

Exactly the same transformation but with different constructions

- 22 Invent your first monad
- 23 More examples of monads
- 24 Monads and their laws
- 25 Program transformations and monads**
- 26 Monads as a general programming technique
- 27 Monads and ML Functors

Commonalities of program transformations

Let us temporarily abandon Haskell and return to pseudo-OCaml syntax

Consider the conversions to exception-returning style, state-passing style, and continuation-passing style. For constants, variables and λ -abstractions (ie., *values*), we have:

Pure	Exceptions	State	Continuations
$\llbracket N \rrbracket$	$= \text{Val}(N)$	$= \lambda s.(N, s)$	$= \lambda k.kN$
$\llbracket x \rrbracket$	$= \text{Val}(x)$	$= \lambda s.(x, s)$	$= \lambda k.kx$
$\llbracket \lambda x.a \rrbracket$	$= \text{Val}(\lambda x.\llbracket a \rrbracket)$	$= \lambda s.(\lambda x.\llbracket a \rrbracket), s)$	$= \lambda k.k(\lambda x.\llbracket a \rrbracket)$

In all three cases we **return** the values N , x , or $\lambda x.\llbracket a \rrbracket$ wrapped in some appropriate context.

For `let` bindings we have

$$\llbracket \text{let } x = a \text{ in } b \rrbracket = \text{match } \llbracket a \rrbracket \text{ with } \text{Exn}(x) \rightarrow \text{Exn}(x) \mid \text{Val}(x) \rightarrow \llbracket b \rrbracket$$

$$\llbracket \text{let } x = a \text{ in } b \rrbracket = \lambda s. \text{match } \llbracket a \rrbracket s \text{ with } (x, s') \rightarrow \llbracket b \rrbracket s'$$

$$\llbracket \text{let } x = a \text{ in } b \rrbracket = \lambda k. \llbracket a \rrbracket (\lambda x. \llbracket b \rrbracket k)$$

In all three cases we extract the value resulting from the computation $\llbracket a \rrbracket$, we **bind** it to the variable x and proceed with the computation $\llbracket b \rrbracket$.

Commonalities of program transformations

For applications we have

$$\begin{aligned} \llbracket ab \rrbracket &= \text{match } \llbracket a \rrbracket \text{ with} \\ &\quad | \text{Exn}(x_a) \rightarrow \text{Exn}(x_a) \\ &\quad | \text{Val}(x_a) \rightarrow \text{match } \llbracket b \rrbracket \text{ with} \\ &\quad\quad | \text{Exn}(y_b) \rightarrow \text{Exn}(y_b) \\ &\quad\quad | \text{Val}(y_b) \rightarrow x_a y_b \end{aligned}$$

$$\begin{aligned} \llbracket ab \rrbracket &= \lambda s. \text{match } \llbracket a \rrbracket s \text{ with } (x_a, s') \rightarrow \\ &\quad \text{match } \llbracket b \rrbracket s' \text{ with } (y_b, s'') \rightarrow x_a y_b s'' \end{aligned}$$

$$\llbracket a b \rrbracket = \lambda k. \llbracket a \rrbracket (\lambda x_a. \llbracket b \rrbracket (\lambda y_b. x_a y_b k))$$

We **bind** the value of $\llbracket a \rrbracket$ to the variable x_a , then **bind** the value of $\llbracket b \rrbracket$ to the variable y_b , then perform the application $x_a y_b$, and rewrap the result as needed.

Commonalities of program transformations

For types notice that if $a : \tau$ then $\llbracket a \rrbracket : \llbracket \tau \rrbracket$ mon

where

- $\llbracket \tau_1 \rightarrow \tau_2 \rrbracket = \tau_1 \rightarrow \llbracket \tau_2 \rrbracket$ mon
- $\llbracket B \rrbracket = B$ for bases types B .

For exceptions:

type α mon = Val of α | Exn of exn

For states:

type α mon = state $\rightarrow \alpha \times$ state

For continuations:

type α mon = ($\alpha \rightarrow$ answer) \rightarrow answer

The previous three translations are instances of the following translation

$$\begin{aligned} \llbracket N \rrbracket &= \text{return } N \\ \llbracket x \rrbracket &= \text{return } x \\ \llbracket \lambda x. a \rrbracket &= \text{return } (\lambda x. \llbracket a \rrbracket) \\ \llbracket \text{let } x = a \text{ in } b \rrbracket &= \llbracket a \rrbracket \gg= (\lambda x. \llbracket b \rrbracket) \\ \llbracket ab \rrbracket &= \llbracket a \rrbracket \gg= (\lambda x_a. \llbracket b \rrbracket \gg= (\lambda y_b. x_a y_b)) \end{aligned}$$

just the monad changes, that is, the definitions of `bind` and `return`).

Exception monad

So the previous translation coincides with our exception returning transformation for the following definitions of bind and return:

```
type  $\alpha$  mon = Val of  $\alpha$  | Exn of exn  
return a      = Val(a)  
m >>= f      = match m with Exn(x) -> Exn(x) | Val(x) -> f x
```

bind encapsulates the propagation of exceptions in compound expressions such as a b or let bindings. As usual we have:

```
return :  $\alpha \rightarrow \alpha$  mon  
(>>=)  :  $\alpha$  mon  $\rightarrow (\alpha \rightarrow \beta$  mon)  $\rightarrow \beta$  mon
```

Additional operations in this monad:

```
raise x = Exn(x)  
trywith m f = match m with Exn(x) -> f x | Val(x) -> Val(x)
```

To have the state-passing transformation we use instead the following definitions for return and bind:

```
type  $\alpha$  mon = state  $\rightarrow$   $\alpha$   $\times$  state
return a =  $\lambda$ s. (a, s)
m >>= f =  $\lambda$ s. match m s with (x, s') -> f x s'
```

bind encapsulates the threading of the state in compound expressions.

Additional operations in this monad:

```
ref x =  $\lambda$ s. store_alloc x s
deref r =  $\lambda$ s. (store_read r s, s)
assign r x =  $\lambda$ s. store_write r x s
```

The Continuation monad

Finally the following monad instance yields the continuation-passing transformation:

```
type  $\alpha$  mon = ( $\alpha \rightarrow$  answer)  $\rightarrow$  answer  
return a =  $\lambda k. k\ a$   
m >>= f =  $\lambda k. m\ (\lambda v. f\ v\ k)$ 
```

Additional operations in this monad:

```
callcc f =  $\lambda k. f\ k\ k$   
throw x y =  $\lambda k. x\ y$ 
```

We can extend the monadic translation to more constructions of the language.

$$\llbracket \mu f. \lambda x. a \rrbracket = \text{return}(\mu f. \lambda x. \llbracket a \rrbracket)$$

$$\llbracket a \text{ op } b \rrbracket = \llbracket a \rrbracket \gg= (\lambda x_a. \llbracket b \rrbracket \gg= (\lambda y_b. \text{return}(x_a \text{ op } y_b)))$$

$$\llbracket C(a_1, \dots, a_n) \rrbracket = \llbracket a_1 \rrbracket \gg= (\lambda x_1. \dots \llbracket a_n \rrbracket \gg= (\lambda x_n. \text{return}(C(x_1, \dots, x_n))))$$

$$\begin{aligned} \llbracket \text{match } a \text{ with } ..p.. \rrbracket &= \llbracket a \rrbracket \gg= (\lambda x_a. \text{match } x_a \text{ with } ..\llbracket p \rrbracket\dots) \\ &\quad \text{where } \llbracket C(x_1, \dots, x_n) \rightarrow a \rrbracket = C(x_1, \dots, x_n) \rightarrow \llbracket a \rrbracket \end{aligned}$$

All these are parametric in the definition of bind and return.

The fundamental property of the monadic translation is that it does not alter the semantics of the computation it encodes. It just adds to the computation some effects.

Theorem

If $a \Rightarrow v$, then $\llbracket a \rrbracket \equiv \text{return } v'$

$$\text{where } v' = \begin{cases} N & \text{if } v = N \\ \lambda x. \llbracket a \rrbracket & \text{if } v = \lambda x. a \end{cases}$$

Examples of monadic translation

```
[[ 1 + f x ]] =  
  (return 1) >>= (λx_1.  
  ((return f) >>= (λx_2.  
    (return x) >>= (λx_3. x_2 x_3)))) >>= ( λx_4.  
  return (x_1 + x_4)))
```

After administrative reductions using the first monadic law:

```
[[ 1 + f x ]] =  
  (f x) >>= (λy. return (1 + y))
```

A second example

```
[[ μfact. λn. if n = 0 then 1 else n * fact(n-1) ]] =  
  return (μfact. λn.  
    if n = 0  
    then return 1  
    else (fact(n-1)) >>= (λv. return (n * v)))
```

What we have done:

- 1 Take a program that performs some computation
- 2 Apply the monadic transformation to it. This yields a new program that uses `return` and `>>=` in it.
- 3 Choose a monad (that is, choose a definition for `return` and `>>=`) and the new programs embeds the computation in the corresponding monad (side-effects, exceptions, etc.)
- 4 You can now add in the program the operations specific to the chosen monad: although it includes effects the program is still *pure*.

- 22 Invent your first monad
- 23 More examples of monads
- 24 Monads and their laws
- 25 Program transformations and monads
- 26 Monads as a general programming technique**
- 27 Monads and ML Functors

Monads as a general programming technique

Monads provide a systematic way to *structure* programs into two well-separated parts:

- the algorithms proper, and
- the “plumbing” of computations needed by these algorithms (state passing, exception handling, non-deterministic choice, etc).

In addition, monads can also be used to *modularize* code and offer new possibilities for reuse:

- Code in monadic form can be parametrized over a monad and reused with several monads.
- Monads themselves can be built in an incremental manner.

Back to Haskell

Let us put all this at work by writing in Haskell the canonical, efficient interpreter that ended our refresher course on operational semantics.

The canonical, efficient interpreter in OCaml (reminder)

```
# type term = Const of int | Var of int | Lam of term
           | App of term * term | Plus of term * term
and value = Vint of int | Vclos of term * environment
and environment = value list (* use Vec instead *)

# exception Error

# let rec eval e a =
  match a with
  | Const n -> Vint n
  | Var n -> List.nth e n
  | Lam a -> Vclos(Lam a, e)
  | App(a, b) -> ( match eval e a with
                  | Vclos(Lam c, e') ->
                    let v = eval e b in
                    eval (v :: e') c
                  | _ -> raise Error)
  | Plus(a,b) -> match (eval e a, eval e b) with
                  | (Vint n, Vint m) -> Vint (n+m)
                  | _ -> raise Error

# eval [] (Plus(Const(5), (App(Lam(Var 0), Const(2))))));; (* 5+((λx.x)2)→7 *)
- : value = Vint 7
```

Note: a Plus operator added

The canonical, efficient interpreter in Haskell

```
data Exp    = Const Integer           -- expressions
            | Var Integer
            | Plus Exp Exp
            | Abs Exp
            | App Exp Exp
data Value  = Vint Integer            -- values
            | Vclos Env Exp
type Env    = [Value]                -- list of values

eval0 :: Env -> Exp -> Value
eval0 env (Const i)    = Vint i
eval0 env (Var n)      = env !! n      -- n-th element
eval0 env (Plus e1 e2) = let Vint i1 = eval0 env e1
                          Vint i2 = eval0 env e2  -- let syntax
                          in Vint (i1 + i2)
eval0 env (Abs e)      = Vclos env e
eval0 env (App e1 e2)  = let Vclos env0 body = eval0 env e1
                          val = eval0 env e2
                          in eval0 (val : env0) body
```

No exceptions: pattern matching may fail.

```
*Main> eval0 [] (App (Const 3) (Const 4))
*** Irrefutable pattern failed for pattern Main.Vclos env body
```

Haskell “do” Notation

Haskell has a very handy notation for monads

In a do block you can macro expand every intermediate line of the form

pattern <- *expression* into *expression* >>= \ *pattern* ->

and every intermediate line of the form

expression into *expression* >>= \ _ ->

This allows us to simplify the monadic translation for expressions which in Haskell syntax is defined as

$\llbracket N \rrbracket = \text{return } N$

$\llbracket x \rrbracket = \text{return } x$

$\llbracket \lambda x. a \rrbracket = \text{return } (\lambda x. \llbracket a \rrbracket)$

$\llbracket \text{let } x = a \text{ in } b \rrbracket = \llbracket a \rrbracket \gg= (\lambda x. \llbracket b \rrbracket)$

$\llbracket ab \rrbracket = \llbracket a \rrbracket \gg= (\lambda x_a. \llbracket b \rrbracket \gg= (\lambda y_b. x_a y_b))$

By using the do notation the last two cases become far simpler to understand

Monadic transformation in Haskell

$$\begin{aligned} \llbracket N \rrbracket &= \text{return } N \\ \llbracket x \rrbracket &= \text{return } x \\ \llbracket \lambda x. a \rrbracket &= \text{return } (\backslash x \rightarrow \llbracket a \rrbracket) \\ \llbracket \text{let } x = a \text{ in } b \rrbracket &= \text{do } x \leftarrow \llbracket a \rrbracket \\ &\quad \llbracket b \rrbracket \\ \llbracket ab \rrbracket &= \text{do } x_a \leftarrow \llbracket a \rrbracket \\ &\quad y_b \leftarrow \llbracket b \rrbracket \\ &\quad x_a y_b \end{aligned}$$

The translation shows that `do` is the monadic version of `let`.

Monad at work

Let us apply the transformation to our canonical efficient interpreter

The canonical, efficient interpreter in monadic form

```
newtype Identity a = MkId a

instance Monad Identity where
    return a      = MkId a           -- i.e. return = id
    (MkId x) >>= f = f x           -- i.e. x >>= f = f x

eval1 :: Env -> Exp -> Identity Value
eval1 env (Const i)    = return (Vint i)
eval1 env (Var n)      = return (env !! n)
eval1 env (Plus e1 e2) = do Vint i1 <- eval1 env e1
                           Vint i2 <- eval1 env e2
                           return (Vint (i1 + i2))
eval1 env (Abs e)      = return (Vclos env e)
eval1 env (App e1 e2)  = do Vclos env0 body <- eval1 env e1
                           val <- eval1 env e2
                           eval1 (val : env0) body
```

We just replaced “do” for “let”, replaced “<-” for “=”, and put “return” in front of every value returned. Let us try to execute $(\lambda x.(x + 1))^4$

```
*Main> let MkId x = (eval1 [] (App(Abs(Plus(Var 0)(Const 1)))(Const 4)))
        in x
Vint 5
```

Although we wrote `eval1` for the Identity monad, the type of `eval1` could be generalized to

```
eval1 :: Monad m => Env -> Exp -> m Value,
```

because we do not use any monadic operations other than `return` and `>>=` (hidden in the `do` notation): `no raise`, `assign`, `trywith`,

Recall that the type

```
Monad m => Env -> Exp -> m Value,
```

reads “for all type `m` that is an instance of the type class `Monad`, the function has type `Env -> Exp -> m Value`”.

In our first definition of `eval1` we explicitly instantiated `m` into the Identity monad, but we can let the system instantiate it. For instance, if we give `eval1` the generalized type above, then we do not need to extract the value encapsulated in the effect:

```
*Main> (eval1 [] (App(Abs(Plus(Var 0)(Const 1)))(Const 4)))  
Vint 5
```

The `ghci` prompt has run the expression in (ie, instantiated `m` by) the IO monad, because internally the interpreter uses the `print` function, which lives in just this monad.

Instantiating eval with the Exception monad

We decide to instantiate `m` in `eval` with the following monad:

```
data Exception e a = Val a | Exn e

instance Monad (Exception e) where
  return x    = Val x
  m >>= f    = case m of
                 Exn x -> Exn x
                 Val x -> f x

raise :: e -> Exception e a
raise x = Exn x

trywith :: Exception e a -> (e -> Exception e a) -> Exception e a
trywith m f = case m of
                Exn x -> f x
                Val x -> Val x
```

Note: Haskell provides an `Error` monad for exceptions. Not dealt with here.

Instantiating eval with the Exception monad

We can do dull instantiation:

```
eval1 :: Env -> Exp -> Exception e Value
eval1 env (Const i )    = return (Vint i)
eval1 env (Var n)       = return (env !! n)
eval1 env (Plus e1 e2 ) = do Vint i1 <- eval1 env e1
                             Vint i2 <- eval1 env e2
                             return (Vint (i1 + i2))
eval1 env (Abs e)       = return (Vclos env e)
eval1 env (App e1 e2 )  = do Vclos env0 body <- eval1 env e1
                             val <- eval1 env e2
                             eval1 (val : env0) body
```

Not interesting since all we obtained is to encapsulate the result into a Val constructor.

The smart way

Use the exception monad to do as the OCaml implementation and raise an error when the applications are not well-typed

Instantiating eval with the Exception monad

New interpreter with exceptions:

```
eval2 :: Env -> Exp -> Exception String Value -- exceptions as strings
eval2 env (Const i) = return (Vint i)
eval2 env (Var n) = return (env !! n)
eval2 env (Plus e1 e2) = do x1 <- eval2 env e1
                             x2 <- eval2 env e2
                             case (x1, x2) of
                               (Vint i1, Vint i2)
                                 -> return (Vint (i1 + i2))
                               _ -> raise "type error in addition"
eval2 env (Abs e) = return (Vclos env e)
eval2 env (App e1 e2) = do fun <- eval2 env e1
                             val <- eval2 env e2
                             case fun of
                               Vclos env0 body
                                 -> eval2 (val : env0) body
                               _ -> raise "type error in application"
```

And we see that the exception is correctly raised

```
*Main> let Val x = ( eval2 [] (App (Abs (Var 0)) (Const 3)) ) in x
Vint 3
*Main> let Exn x = ( eval2 [] (App (Const 2) (Const 3)) ) in x
"type error in application"
```

Instantiating `eval` with the `Exception` monad

Advantages:

- The function `eval2` is *pure*!
- Module few syntactic differences the code is really the same as code that would be written in an impure language (*cf.* the corresponding OCaml code)
- All “plumbing” necessary to preserve purity is defined separately (eg, in the `Exception` monad and its extra functions)
- In most cases the programmer does not even need to define “plumbing” since monads provided by standard Haskell libraries are largely sufficient.

A second try

Let us instantiate the type `Monad m => Env -> Exp -> m Value` with a different monad `m`. For our next example we choose the `State` monad.

Instantiating eval with the State monad

Goal: Add profiling capabilities by recording the number of evaluation steps.

```
-- The State Monad
data State s a = MkSt (s -> (a,s))

instance Monad (State s) where
  return a      = MkSt (\s -> (a,s))
  (MkSt g) >>= f = MkSt (\s -> let (v,s') = g s
                                MkSt h = f v
                                in h s')
```

```
get :: State s s
get = MkSt (\s -> (s,s))

put :: s -> State s ()
put s = MkSt (\_ -> ((),s))
```

To count evaluation steps we use an Integer number as state (ie, we use the `State Integer` monad). The operation `tick`, retrieves the hidden state from the computation, increases it and stores it back

```
tick :: State Integer ()
tick = do st <- get
         put (st + 1)
```

Instantiating eval with the State monad

```
eval3 :: Env -> Exp -> State Integer Value
eval3 env (Const i)    = do tick
                          return (Vint i)
eval3 env (Var n)      = do tick
                          return (env !! n)
eval3 env (Plus e1 e2) = do tick
                          x1 <- eval3 env e1
                          x2 <- eval3 env e2
                          case (x1 , x2) of
                            (Vint i1, Vint i2)
                              -> return (Vint (i1 + i2 ))
eval3 env (Abs e)      = do tick
                          return (Vclos env e)
eval3 env (App e1 e2)  = do tick
                          fun <- eval3 env e1
                          val <- eval3 env e2
                          case fun of
                            Vclos env0 body
                              -> eval3 (val : env0 ) body
```

The evaluation of $(\lambda x.x)3$ takes 4 steps of reduction. This is shown by giving 0 as initial value of the state:

```
*Main> let MkSt s = eval3 [] (App (Abs (Var 0)) (Const 3)) in s 0
(Vint 3,4)
```

Combining monads the hard way

What if we want *both* exceptions and state in our interpreter?

- Merging the code of `eval2` and `eval3` is straightforward: just add the code of `eval2` that raises the type-error exceptions at the end of the `Plus` and `App` cases in the definition of `eval3`.
- The problem is how to define the monad that support both effects.

We can *write from scratch* the monad `m` that supports both effects.

```
eval4 :: Monad m => Env -> Exp -> m Value
```

Where the monad `m` above is one of the following two cases:

- 1 Use `StateOfException s` for `m`:

```
data StateOfException s e a = State (s -> Exception e (s,a))
```

the computation can either return a new pair state, value or generate an error (ie, when an exception is raised the state is discarded)

- 2 Use `ExceptionOfState s` for `m`:

```
data ExceptionOfState s e a = State (s -> ((Exception e a), s))
```

the computation always returns a pair value and new state, and the value in this pair can be either an error or a normal value.

Combining monads the hard way

Notice that for the case `State (s -> ((Exception e a), s))` there are two further possibilities, according to the state we return when an exception is caught. Each possibility corresponds to a different definition of `trywith`

- 1 backtrack the modifications made by the computation `m` that raised the exception:

```
trywith m f = \s -> case m s of
    (Val x , s') -> (Val x , s')
    (Exn x , s') -> f x s
```

- 2 keep the modifications made by the computation `m` that raised the exception:

```
trywith m f = \s -> case m s of
    (Val x , s') -> (Val x , s')
    (Exn x , s') -> f x s'
```

Avoid the boilerplate

Each of the standard monads is specialised to do exactly one thing. In real code, we often need several effects at once. Composing monads by hand or rewriting them from scratch soon reaches its limits

Combining monads by **compositionality**

By applying the monadic transformation to `eval` we passed from a function of type

```
Env -> Exp -> Value,
```

to a function of type

```
Monad m => Env -> Exp -> m Value,
```

In this way we made the code for `eval` parametric in the monad `m`.

Later we chose to instantiate `m` to some particular monad in order to use the specific characteristics

IDEA: transform the code of an `instance` definition of the monad class so that this definition becomes parametric in some other monad `m`.

Monad transformer

A monad instance that is parametric in another monad is a *monad transformer*.

To work on the monad parameter, apply the monadic transformation to the definitions of instances

Monad Transformers can help:

- A monad transformer transforms a monad by adding support for an additional effect.
- A library of monad transformers can be developed, each adding a specific effect (state, error, . . .), allowing the programmer to mix and match.
- A form of *aspect-oriented programming*.

Monad Transformation in Haskell

- A *monad transformer* maps monads to monads. Represented by a type constructor `T` of the following kind:

$$T :: (* -> *) -> (* -> *)$$

- Additionally, a monad transformer adds computational effects. A mapping `lift` from computations in the underlying monad to computations in the transformed monad is needed:

$$\text{lift} :: M a -> T M a$$

Are you lost? ... Let us recap

Goal: write the following code where all the **plumbing** to handle effects is hidden in the definition of **m**

```
eval :: (Monad m) => Env -> Exp -> m Value

eval env (Const i )   = do tick
                        return (Vint i)
eval env (Var n)      = do tick
                        return (env !! n)
eval env (Plus e1 e2) = do tick
                        x1 <- eval env e1
                        x2 <- eval env e2
                        case (x1 , x2) of
                            (Vint i1, Vint i2)
                                -> return (Vint (i1 + i2 ))
                            _      -> raise "type error in addition"

eval env (Abs e)      = do tick
                        return (Vclos env e)
eval env (App e1 e2)  = do tick
                        fun <- eval env e1
                        val <- eval env e2
                        case fun of
                            Vclos env0 body
                                -> eval (val : env0 ) body
                            _      -> raise "type error in application"
```

Are you lost? ... Let us recap

The *dirty work* is in the definition of the monad `m` that will be used. Two ways are possible:

- 1 **Define `m` from scratch:** Define a new monad `m` so as it combines the effects of the Exception and of the State monads for which `raise` and `tick` are defined.

Advantages: a fine control on the definition

Drawbacks: no code reuse, hard to maintain and modify

- 2 **Define `m` by composition:** Define `m` by composing more elementary blocks that provide functionalities of states and exceptions respectively
- Advantages:** modular development; in many case it is possible to reuse components from the shelves.

Drawbacks: Some trade-off since the building blocks may not provide exactly the sought combination of functionalities.

Monad transformers

We show the second technique by building the sought `m` from two *monad transformers* for exceptions and states respectively.

Step 1: defining the functionalities

We define two subclasses of the Monad class

EXCEPTION MONAD

An Exception Monad is a monad with an operation `raise` that takes a string and yields a monadic computation

```
class Monad m => ExMonad m where
  raise :: String -> m a
```

STATE MONAD

A State Monad is a monad with an operation `tick` that yields a computation on values of the unit type.

```
class Monad m => StMonad m where
  tick :: m ()
```

It is now possible to specify a type for `eval` so that its definition type-checks

```
eval :: (ExMonad m, StMonad m) => Env -> Exp -> m Value
eval env (Const i) = do tick
                       ⋮
                       -> raise "type error in addition"
```

Step 2: defining the building blocks

We now need to define a monad `m` that is an instance of both `StMonad` and `ExMonad`.

We do it by composing two *monad transformers*

Definition (Monad transformer)

A *monad transformer* is a higher-order operator `t` that maps each monad `m` to a monad `(t m)`, equipped with an operation `lift` that promotes a computation `x :: m a` from the original monad `m` that is fed to `t` to a computation

$$(\text{lift } x) :: (t \ m) \ a$$

on the monad `(t m)`.

Definition of the class of monad transformers

```
class MonadTrans t where
  lift :: Monad m => m a -> (t m) a
```

Example

If we want to apply to the monad `Exception String` a transformer `T` that provides some operation `xyz`, then we need to lift `raise` from `Exception String` to `T(Exception String)`.

Without the lifting the only operation defined for `T(Exception String)` would be `xyz`. With `lift` since

```
raise :: String -> Exception String,
```

then:

```
lift.raise :: String -> T(Exception String)
```

Nota bene

There is no magic formula to produce the transformer versions of a given monad

Step 2a: A monad transformer for exceptions

Consider again our first monad `Exception e`:

```
data Exception e a = Val a | Exn e

instance Monad (Exception e) where
  return x    = Val x
  m >>= f    = case m of Exn x -> Exn x ; Val x -> f x

raise :: e -> Exception e a
raise x = Exn x
```

We now want to modify the code above in order to obtain a transformer `ExceptionT` in which the computations are themselves on monads, that is:

```
data ExceptionT m a = MkExc (m (Exception String a))
```

The (binary) type constructor `ExceptionT` “puts exceptions inside” another monad `m` (convention: a monad transformer is usually named as the corresponding monad with a ‘T’ at the end.)

We want `ExceptionT` to be a *monad transformer*, ie. `(ExceptionT m)` to be a monad: we must define the bind and return for the monad `(ExceptionT m)`:

```

-- The 'recover' function just strips off the outer MkExc constructor,
-- for convenience
recover :: ExceptionT m a -> m (Exception String a)
recover (MkExc x) = x

-- return is easy. It just wraps the value first in the monad m
-- by return (of the underlying monad) and then in MkExc

returnET :: (Monad m) => a -> ExceptionT m a
returnET x = MkExc (return (Val x))

-- A first version for bind uses do and return to work on the
-- underlying monad m ... whatever it is.

bindET :: (Monad m) => (ExceptionT m a) -> ( a -> ExceptionT m b)
                                         -> ExceptionT m b
bindET (MkExc x) f =                    -- x is of type m (Exception a)
    MkExc (                               -- we wrap the result in MkExc
        do y <- x                         -- y is of type Exception a
           case y of
             Val z -> recover (f z)
             Exn z -> return (Exn z) )

```

Notice the use of the monadic syntax (`do`, `return`, ...) to work on the monad parameter `m`.

Step 2a: A monad transformer for exceptions

More compactly:

```
instance Monad m => Monad (ExceptionT m) where
  return x = MkExc (return (Val x))
  x >>= f = MkExc (recover x >>= r)
             where r (Exn y) = return (Exn y)
                   r (Val y) = recover (f y)
```

Moreover, (`ExceptionT m`) is an exception monad, not just a plain one...

```
instance Monad m => ExMonad (ExceptionT m) where
  raise e = MkExc (return (Exn e))
```

`ExceptionT` is a monad transformer because we can lift any action in `m` to an action in (`ExceptionT m`) by wrapping its result in a 'Val' constructor...

```
instance MonadTrans ExceptionT where
  lift g = MkExc $ do { x <- g; return (Val x) }
```

We can now use the `lift` operation to make (`ExceptionT m`) into a state monad whenever `m` is one, by lifting `m`'s `tick` operation to (`ExceptionT m`).

```
instance StMonad m => StMonad (ExceptionT m) where
  tick = lift tick
```

Step 2b: A monad transformer for states

```
newtype StateT m a = MkStt ( Int -> m (a,Int))

-- strip off the MkStt constructor
apply :: StateT m a -> Int -> m (a, Int)
apply (MkStt f) = f

-- if m is a monad, then StateT m is a monad
instance Monad m => Monad (StateT m) where
  return x = MkStt $ \s -> return (x,s)
  p >>= q = MkStt $ \s -> do (x,s') <- apply p s
                             apply (q x) s'

-- StateT is a monad transformer
instance MonadTrans StateT where
  lift g = MkStt $ \s -> do x <- g; return (x,s)

-- if m is a monad, then StateT m is not only a monad
-- but also a STATE MONAD
instance (Monad m) => StMonad (StateT m) where
  tick = MkStt $ \s -> return ((), s+1)

-- use lift to promote StateT m to an exception monad
instance ExMonad m => ExMonad (StateT m) where
  raise e = lift (raise e)
```

Lost again? Let us recap this Step 2

In Step 2 we defined some monad transformers of the form `XYZT`.

- 1 To be a “transformer” `XYZT` must map monads into monads. So if `m` is a monad (ie., it provides `bind` and `return`), then so must `(XYZT m)` be. So we define `bind` and `return` for `(XYZT m)` and use monadic notation to work on the generic `m`.
- 2 But `(XYZT m)` must not only provide `bind` and `return`, but also some operations typical of some class `XYZ`, subclass of the `Monad` class. So we define also these operations by declaring that `(XYZT m)` is an instance of `XYZ`.
- 3 This is not enough for `XYZT` to be a transformer. It must also provide a `lift` operation. By defining it we declare that `XYZT` is an instance of the class `MonadTrans`
- 4 Finally we can use the `lift` function to make `(XYZT m)` “inherit” the characteristics of `m`: so if `m` is an instance of some monadic subclass `Abc`, then we can make also `(XYZT m)` be a `Abc` monad simply by lifting all the operations specific of `Abc`.

Step 3: Putting it all together...

Just a matter of assembling the pieces.

Interestingly, though, there are TWO ways to combine our transformers to build a monad with exceptions and state:

```
1 evalStEx :: Env -> Exp -> StateT (ExceptionT Identity) Value
  evalStEx = eval
```

```
2 evalExSt :: Env -> Exp -> ExceptionT (StateT Identity) Value
  evalExSt = eval
```

Note that `ExceptionT Identity` and `StateT Identity` are respectively the `Exception` and `State` monads defined before, modulo two modifications:

- 1 Values are further wrapped in an inner `MkId` constructor
- 2 To enhance readability I used distinct names for the types and their constructors, for instance:

```
newtype StateT m a = MkStt (Int -> m (a,Int))
```

rather than

```
newtype StateT m a = StateT (Int -> m (a,Int))
```

as it is custom in the Haskell library

Order matters

At first glance, it appears that `evalExSt` and `evalStEx` do the same thing...

```
five  = (App(Abs(Plus(Var 0)(Const 1)))(Const 4))    --(λx.(x+1))4
wrong = (App(Abs(Plus(Var 0)(Const 1)))(Abs(Var 0))) --(λx.(x+1))(λy.y)

*Main> evalStEx [] five
Vint 5, count: 6

*Main> evalExSt [] five
Vint 5, count: 6
```

BUT ...

```
*Main> evalStEx [] wrong
exception: type error in addition

*Main> evalExSt [] wrong
exception: type error in addition, count: 6
```

- `StateT (ExceptionT Identity)` either returns a state or an exception
- `ExceptionT (StateT Identity)` always returns a state

I omitted the code to print the results of monadic computations. It can be found in the accompanying code:
<http://www.pps.jussieu.fr/~gc/slides/evaluator.hs>

The Continuation monad

Computation type: Computations which can be interrupted and resumed.

Binding strategy: Binding a function to a monadic value creates a new continuation which uses the function as the continuation of the monadic computation.

Useful for: Complex control structures, error handling and creating co-routines.

From `haskell.org`:

**Abuse of the Continuation monad can
produce code that is impossible to
understand and maintain.**

Many algorithms which require continuations in other languages do not require them in Haskell, due to Haskell's lazy semantics.

The Continuation monad

```
newtype Cont r a = Cont (a -> r) -> r

instance Monad (Cont r) where
  return a      = Cont $ \k -> k a           -- =  $\lambda k.k a$ 
  (Cont c) >>= f = Cont $ \k -> c (\a -> (f a) k) -- =  $\lambda k.c(\lambda a.f a k)$ 
```

`Cont r a` is a CPS computation that produces an intermediate result of type `a` within a CPS computation whose final result type is `r`.

The return function simply creates a continuation which passes the value on.

The `>>=` operator adds the bound function into the continuation chain.

```
class (Monad m) => MonadCont m where
  callCC :: ((a -> m b) -> m a) -> m a

instance MonadCont (Cont r) where
  callCC f = Cont (\k -> f (\a -> Cont (\_ -> k a)) k)
```

No need to define `throw` since we can directly use the continuation by applying it to a value, as shown in the next example

```
bar :: Char -> String -> Cont r Int
bar c s = do
  msg <- callCC $ \k -> do
    let s' = c : s
        if (s' == "hello") then k "They say hello." else return ()
    let s'' = show s'
    return ("They appear to be saying " ++ s'')
  return (length msg)
```

When you call `k` with a value, the entire `callCC` call takes that value. In other words, `k` is a bit like a 'goto' statement in other languages: when we call `k` in our example, it pops the execution out to where you first called `callCC`, the `msg <- callCC $...` line. No more of the argument to `callCC` (the inner `do`-block) is executed. Hence the following example contains a useless line:

```
bar :: Cont r Int
bar = callCC $ \k -> do
  let n = 5
  k n
  return 25
```

`bar` will always return `5`, and never `25`, because we pop out of `bar` before getting to the `return 25` line.

Summary

Purity has advantages but effects are unavoidable.

- To have them both, effects must be explicitly programmed.
- In order to separate the definition of the algorithm from the definition of the plumbing that manage the effects it is possible to use a monad. This centralizes all the programming that concerns effects.
- Several effects may be necessary in the same program. One can define the corresponding monad by composing monad transformers. These are functions from monads to monads, each handling a specific effect.

However

- Putting code in monadic form is easy and can be done automatically, but there is no magic formula to define monads or even derive the corresponding transformers
- Understanding monadic code is relatively straightforward but writing and debugging monads or monads transformers from scratch may be dreadful.

Suggestion

Use **existing** monads and monads transformers as much as you can.

- 22 Invent your first monad
- 23 More examples of monads
- 24 Monads and their laws
- 25 Program transformations and monads
- 26 Monads as a general programming technique
- 27 Monads and ML Functors**

- Monads define the `bind` and `return` functions that are the core of the plumbing of effects
- Specific operations for effects such as `raise` and `tick` are provided by subclasses of Monads (eg, `StMonad`, `ExMonad`).
- Modular development is obtained by *monad transformers* which are functions from monads to (subclasses of) monads.

We can reproduce monads by modules and transformers by functors.

The Caml module signature for a monad is:

```
module type MONAD = sig
  type  $\alpha$  mon
  val return:  $\alpha$  ->  $\alpha$  mon
  val bind:  $\alpha$  mon -> ( $\alpha$  ->  $\beta$  mon) ->  $\beta$  mon
end
```

The Identity monad

The Identity monad is a trivial instance of this signature:

```
module Identity = struct
  type  $\alpha$  mon =  $\alpha$ 
  let return x = x
  let bind m f = f m
end
```

Monad transformer for exceptions

```
module ExceptionT(M: MONAD) = struct
  type  $\alpha$  outcome = Val of  $\alpha$  | Exn of exn
  type  $\alpha$  mon = ( $\alpha$  outcome) M.mon
  let return x = M.return (Val x)
  let bind m f =
    M.bind m (function Exn e -> M.return (Exn e) | Val v -> f v)
  let lift x = M.bind x (fun v -> M.return (Val v))
  let raise e = M.return (Exn e)
  let trywith m f =
    M.bind m (function Exn e -> f e | Val v -> M.return (Val v))
end
```

Notice the lesser flexibility due to the lack of overloading which obliges us to specify whose `bind` and `return` we use.

Monad transformer for state

```
module StateT(M: MONAD) = struct
  type  $\alpha$  mon = state -> ( $\alpha$  * state) M.mon
  let return x = fun s -> M.return (x, s)
  let bind m f =
    fun s -> M.bind (m s) (fun (x, s') -> f x s')
  let lift m = fun s -> M.bind m (fun x -> M.return (x, s))
  let ref x = fun s -> M.return (store_alloc x s)
  let deref r = fun s -> M.return (store_read r s, s)
  let assign r x = fun s -> M.return (store_write r x s)
end
```

Using monad transformers

```
module State = StateT(Identity)

module StateAndException = struct
  include ExceptionT(State)
  let ref x = lift (State.ref x)
  let deref r = lift (State.deref r)
  let assign r x = lift (State.assign r x)
end
```

This gives a type $\alpha \text{ mon} = \text{state} \rightarrow \alpha \text{ outcome} \times \text{state}$, i.e. state is preserved when raising exceptions. The other combination, `StateT(ExceptionT(Identity))` gives $\alpha \text{ mon} = \text{state} \rightarrow (\alpha \times \text{state}) \text{ outcome}$, i.e. state is discarded when an exception is raised.

Exercise

Define the functor for continuation monad transformer.

```
module ContTransf(M: MONAD) = struct
  type  $\alpha$  mon = ( $\alpha$  -> answer M.mon) -> answer M.mon
  let return x =
  let bind m f =
  let lift m =

  let callcc f =
  let throw c x =
end
```

- Philip Wadler. Monads for functional Programming. In Advanced Functional Programming, Proceedings of the Baastad Spring School, Lecture Notes in Computer Science n. 925, Springer, 1995.
- Martin Grabmüller. Monad Transformers Step by Step, Unpublished draft. 2006 <http://www.grabmuelleer.de/martin/www/pub/>